L2 MMAG - Réponses à l'exercice 4 de la série 3 avec Sagemath

F-X. Dehon - (/cdn-cgi/l/email-protection)

Sagemath (http://www.sagemath.org/) est un logiciel de calcul formel qui permet entre autre :

- de faire des calculs algébriques avec les vecteurs (on peut par exemple mettre en oeuvre l'algorithme de Gram-Schmidt pas à pas)
- d'expliciter une base de l'espace des solutions d'un système d'équations linéaires à partir de la matrice de ce système.

On peut exécuter un code Sagemath (une suite d'instructions) sur le site SageMathCell (https://sagecell.sagemath.org/) ou bien en ouvrant un compte et en créant une feuille de calcul sur Cocalc (https://cocalc.com/).

L'exercice en question. Soient $v_1 = (2,1,2,-4)$, $v_2 = (1,2,-2,4)$, $v_3 = (4,2,1,2)$ et $V = \mathrm{vect}(v_1, v_2, v_3) \subset \mathbb{R}^4$. Déterminer $V^\perp$, une base orthonormée de $V^\perp$ et une base orthonormée de V.

1ère méthode : $V^\perp$ est donné par le système d'équations linéaires $< v_i, (x,y,z,t) >= 0,\ i = 1,2,3.$ La matrice de ce système est la matrice dont les lignes sont les coordonnées des v_i.

Dans Sagemath :

- `M=matrix(QQ,[v1,v2,v3])` définit M comme cette matrice à coefficients dans $\mathbb{Q}$,
- `K=M.right_kernel()` définit K comme $\ker(M)$, l'espace des solutions du système $MX = 0$,
- `g=K.basis()` définit g comme la liste des vecteurs d'une base de K déterminée par Sagemath
- `g1=g[0];print g1*g1` définit g_1 comme le premier élément de cette liste et calcule $||g_1||^2 =< g_1, g_1 >$.

In [1]:
```
v1=vector([2,1,2,-4])
v2=vector([1,2,-2,4])
v3=vector([4,2,1,2])
```

In [19]:
```
M=matrix(QQ,[v1,v2,v3]) #latex(matrix(QQ,[v1,v2,v3]))
Vorthog=M.right_kernel()
print Vorthog
g=Vorthog.basis()[0]
print 'g=',g,'|g|^2=', g*g
```

```
Vector space of degree 4 and dimension 1 over Rational Field
Basis matrix:
[   1    -1 -5/4 -3/8]
g= (1, -1, -5/4, -3/8) |g|^2= 237/64
```

$V^\perp$ est de dimension 1 donc un seul vecteur non nul de $V^\perp$ donne une base orthogonale de $V^\perp$. Comme $V^\perp$ est de dimension 1, V est de dimension 3 et (v_1, v_2, v_3) en est une base.

Pour calculer une base orthogonale de V on part d'une base de V, ici (v_1, v_2, v_3) ; on pose $f_1 = v_1$; on cherche les combinaisons linéaires de v_1, v_2, v_3 qui sont orthogonales à f_1 en résolvant l'équation $< f_1, xv_1 + yv_2 + zv_3 >= 0$; on choisit pour f_2 une combinaison non triviale ; on cherche les combinaisons linéaires de v_1, v_2, v_3 qui sont orthogonales à f_1, f_2 en résolvant le système $< f_i, xv_1 + yv_2 + zv_3 >= 0,\ i = 1,2$; on choisit pour f_3 une combinaison non triviale. La matrice d'un tel système est le produit de la matrice dont les lignes sont les coordonnées des f_i avec la matrice dont les colonnes sont les coordonnées de v_1, v_2, v_3.

In [2]:
```
M=matrix(QQ,[v1,v2,v3]).transpose()
f1=v1
V1=(matrix(QQ,f1)*M).right_kernel()
f2=M*V1.basis()[0]
V2=(matrix(QQ,[f1,f2])*M).right_kernel()
f3=M*V2.basis()[0]
show(f1,f2,f3)
show([f*f for f in [f1,f2,f3]])
print f1*f2,f1*f3,f2*f3
```

Out[2]: $$(2,\ 1,\ 2,\ -4)\ \left(-23,\ -\frac{23}{2},\ -\frac{17}{4},\ -\frac{33}{2}\right)\ \left(-\frac{45}{152},\ \frac{141}{76},\ -\frac{30}{19},\ -\frac{9}{19}\right)$$

Out[2]: $$\left[25,\ \frac{15225}{16},\ \frac{144333}{23104}\right]$$

```
0 0 0
```

2ème méthode : Orthogonalisation de Gram-Schmidt.

On commence par orthogonaliser la famille (v_1, v_2, v_3) en posant

$f_1 = v_1$,

$f_2 = v_2 - \mathrm{proj}_{<f_1>}(v_2) = v_2 - \frac{<f_1,v_2>}{<f_1,f_1>} f_1$,

$f_3 = v_3 - \mathrm{proj}_{<f_1,f_2>}(v_3) = v_3 - \left(\frac{<f_1,v_3>}{<f_1,f_1>} f_1 + \frac{<f_2,v_3>}{<f_2,f_2>} f_2\right)$.

In [4]:
```
f1=v1
print 'f1=',f1,',','|f1|^2=',f1*f1
f2=v2-f1*v2/(f1*f1)*f1
print 'f2=',f2,',','|f2|^2=',f2*f2
f3=v3-f1*v3/(f1*f1)*f1-f2*v3/(f2*f2)*f2
print 'f3=',f3,',','|f3|^2=',f3*f3
print 'f1*f2,f1*f3,f2*f3=',f1*f2,f1*f3,f2*f3
```

```
f1= (2, 1, 2, -4) , |f1|^2= 25
f2= (57/25, 66/25, -18/25, 36/25) , |f2|^2= 369/25
f3= (46/41, -46/41, 61/41, 42/41) , |f3|^2= 237/41
f1*f2,f1*f3,f2*f3= 0 0 0
```

f_1, f_2, f_3 sont orthogonaux, non nuls et engendrent $V =< v_1, v_2, v_3 >$ donc V est de dimension 3 et $V^\perp$ est de dimension 1. On a $V^\perp =< g >$ où g complète f_1, f_2, f_3 en une base orthogonale de $\mathbb{R}^4$.

Complétion en une base orthogonale de $\mathbb{R}^4$: On cherche v qui ne soit pas dans V et alors $g = v - \mathrm{proj}_{<v_1,v_2,v_3>}(v)$ est non nul et orthogonal à v_1, v_2, v_3. On essaie successivement pour v les éléments de la base canonique de $\mathbb{R}^4$; l'un d'eux convient puisque $\mathbb{R}^4$ n'est pas inclus dans V.

In [30]:
```
v=vector([1,0,0,0])
g=v-f1*v/(f1*f1)*f1-f2*v/(f2*f2)*f2-f3*v/(f3*f3)*f3
print 'g=',g,',','|g|^2=',g*g
```

```
g= (64/237, -64/237, -80/237, -8/79) , |g|^2= 64/237
```